Euler-Heisenberg matching at one loop

Julian Eicher (julian.eicher@unibe.ch), April 2024

Preliminaries

- Load FeynCalc

In[ ]:=
```
<< FeynCalc`
```

FeynCalc 10.0.0 (stable version). For help, use the online documentation, visit the forum and have a look at the supplied examples.

If you use FeynCalc in your research, please evaluate FeynCalcHowToCite[] to learn how to cite this software.

Please keep in mind that the proper academic attribution of our work is crucial to ensure the future development of this package!

- Use “StandardForm” instead of FeynCalc’s default “TraditionalForm” output format

In[ ]:=
```
SetOptions[EvaluationNotebook[],
 CommonDefaultFormatTypes →
  {"Output" → StandardForm}]
```

- Define kinematics (SPD stands for “scalar product” in d dimensions)

```
In[•]:= kinematics = {SPD[p1, p1] → 0, SPD[p2, p2] → 0,
          SPD[p1, p2] → s / 2}
```

$$\text{Out}[\circ]= \left\{\text{SPD}[p1, p1] \to 0, \text{SPD}[p2, p2] \to 0, \text{SPD}[p1, p2] \to \frac{s}{2}\right\}$$

- Global prefactor (factor (-1) from the fermion trace)

```
In[•]:= prefactor = -e ^ 4
```

$$\text{Out}[\circ]= -e^4$$

- Loop integral formula

```
In[•]:= master[n_] := I * (-1) ^ n / (4 * Pi) ^ (D / 2) / Gamma[n] *
          Gamma[n - D / 2] / (m ^ 2) ^ (n - D / 2)
```

- Write a generic propagator as $\frac{1}{\left((k+p)^2-m^2\right)} = \frac{1}{\Delta + 2\,k\cdot p + p\cdot p}$, where $\Delta = k^2 - m^2$

```
In[•]:= Prop[k_, p_] := 1 / (Δ + 2 * SPD[k, p] + SPD[p, p]) //.
          {SPD[a_, b_ + c_] → SPD[a, b] + SPD[a, c],
           SPD[-a_, b_] → -SPD[a, b]}
```

- In the following, we calculate the contracted forward amplitudes $A_1 = g_{\mu_1 \mu_2}\, g_{\nu_1 \nu_2}\, B^{\mu_1 \mu_2 \nu_1 \nu_2}$ and $A_2 = g_{\mu_1 \nu_1}\, g_{\mu_2 \nu_2}\, B^{\mu_1 \mu_2 \nu_1 \nu_2}$ for each of the three diagrams separately, and then combine the results

Diagram 1



- Numerator (GSD[k] = $\gamma \cdot k$ in d dimensions, GAD[μ] = γ^μ in d dimensions)

```
In[•]:= num1 =
  DiracTrace[(GSD[k] + m).GAD[μ1].(GSD[k - p1] + m).
     GAD[ν1].(GSD[k] + m).GAD[ν2].(GSD[k + p2] + m).
     GAD[μ2]] // DiracSimplify // FCE;
```

- Contraction A_1

```
In[•]:= MTD[μ1, μ2] * MTD[ν1, ν2] * num1 // Contract // FCE;
        num1A1 = % /. kinematics
```

$$\text{Out}[\circ]= 4\,D^2\,m^4 - 2\,D^2\,m^2\,s + 16\,m^2\,\text{SPD}[k,k] + 16\,D\,m^2\,\text{SPD}[k,k] - 8\,D^2\,m^2\,\text{SPD}[k,k] + 8\,s\,\text{SPD}[k,k] - 8\,D\,s\,\text{SPD}[k,k] + 2\,D^2\,s\,\text{SPD}[k,k] + 16\,\text{SPD}[k,k]^2 - 16\,D\,\text{SPD}[k,k]^2 + 4\,D^2\,\text{SPD}[k,k]^2 - 16\,D\,m^2\,\text{SPD}[k,p1] + 4\,D^2\,m^2\,\text{SPD}[k,p1] - 16\,\text{SPD}[k,k]\times\text{SPD}[k,p1] + 16\,D\,\text{SPD}[k,k]\times\text{SPD}[k,p1] - 4\,D^2\,\text{SPD}[k,k]\times\text{SPD}[k,p1] + 16\,D\,m^2\,\text{SPD}[k,p2] - 4\,D^2\,m^2\,\text{SPD}[k,p2] + 16\,\text{SPD}[k,k]\times\text{SPD}[k,p2] - 16\,D\,\text{SPD}[k,k]\times\text{SPD}[k,p2] + 4\,D^2\,\text{SPD}[k,k]\times\text{SPD}[k,p2] - 32\,\text{SPD}[k,p1]\times\text{SPD}[k,p2] + 32\,D\,\text{SPD}[k,p1]\times\text{SPD}[k,p2] - 8\,D^2\,\text{SPD}[k,p1]\times\text{SPD}[k,p2]$$

- Contraction A_2

```
In[•]:= MTD[μ1, ν1] * MTD[μ2, ν2] * num1 // Contract // FCE;
        num1A2 = % /. kinematics
```

$$\text{Out}[\circ]= 4\,D^2\,m^4 - 8\,m^2\,s + 8\,D\,m^2\,s - 2\,D^2\,m^2\,s + 16\,m^2\,\text{SPD}[k,k] + 16\,D\,m^2\,\text{SPD}[k,k] - 8\,D^2\,m^2\,\text{SPD}[k,k] + 8\,s\,\text{SPD}[k,k] - 8\,D\,s\,\text{SPD}[k,k] + 2\,D^2\,s\,\text{SPD}[k,k] + 16\,\text{SPD}[k,k]^2 - 16\,D\,\text{SPD}[k,k]^2 + 4\,D^2\,\text{SPD}[k,k]^2 - 16\,m^2\,\text{SPD}[k,p1] + 4\,D^2\,m^2\,\text{SPD}[k,p1] - 16\,\text{SPD}[k,k]\times\text{SPD}[k,p1] + 16\,D\,\text{SPD}[k,k]\times\text{SPD}[k,p1] - 4\,D^2\,\text{SPD}[k,k]\times\text{SPD}[k,p1] + 16\,m^2\,\text{SPD}[k,p2] - 4\,D^2\,m^2\,\text{SPD}[k,p2] + 16\,\text{SPD}[k,k]\times\text{SPD}[k,p2] - 16\,D\,\text{SPD}[k,k]\times\text{SPD}[k,p2] + 4\,D^2\,\text{SPD}[k,k]\times\text{SPD}[k,p2] - 32\,\text{SPD}[k,p1]\times\text{SPD}[k,p2] + 32\,D\,\text{SPD}[k,p1]\times\text{SPD}[k,p2] - 8\,D^2\,\text{SPD}[k,p1]\times\text{SPD}[k,p2]$$

- Denominator

```
In[ ]:= denom1 =
         Prop[k, 0] * Prop[k, -p1] * Prop[k, 0] * Prop[k, p2] /.
          kinematics
```

$$\text{Out[ ]=} \quad \frac{1}{\Delta^2\,(\Delta - 2\,\text{SPD}[k,\,p1])\,(\Delta + 2\,\text{SPD}[k,\,p2])}$$

- Combine numerator and denominator, and expand in external momenta (use *x* as an auxiliary expansion parameter)

```
In[ ]:= num1A1 * denom1 /. {p1 → x * p1, p2 → x * p2, s → x ^ 2 * s} //.
          SPD[a_, x * b_] → x * SPD[a, b];
        Series[%, {x, 0, 4}] // Normal;
        diag1A1 = % /. x → 1 // Expand;
```

```
In[ ]:= num1A2 * denom1 /. {p1 → x * p1, p2 → x * p2, s → x ^ 2 * s} //.
          SPD[a_, x * b_] → x * SPD[a, b];
        Series[%, {x, 0, 4}] // Normal;
        diag1A2 = % /. x → 1 // Expand;
```

Diagram 2

Diagram 3

Combine diagrams

- Add diagrams

```
In[ ]:= A1full = diag1A1 + diag2A1 + diag3A1;
        A2full = diag1A2 + diag2A2 + diag3A2;
```

- Tensor reduction (loop integral numerator reduction) using TID. Here FAD[{k,m}] = $\frac{1}{k^2-m^2}$ stands for “Feynman amplitude denominator”; TID needs to know what Δ is

In[•]:=
```
A1full /. Δ → 1 / FAD[{k, m}];
TID[%, k] // FeynAmpDenominatorSplit // FCE;
A1red = % /. kinematics /. FAD[{k, m}] → 1 / Δ // Expand
```

Out[•]=

$$\frac{8960\, m^8\, s^2}{D\,(2+D)\,\triangle^8}-\frac{2752\, m^6\, s^2}{(2+D)\,\triangle^7}+\frac{21568\, m^6\, s^2}{D\,(2+D)\,\triangle^7}-\frac{640\, m^6\, s}{(2+D)\,\triangle^6}-\frac{1280\, m^6\, s}{D\,(2+D)\,\triangle^6}-\frac{5248\, m^4\, s^2}{(2+D)\,\triangle^6}+\frac{17344\, m^4\, s^2}{D\,(2+D)\,\triangle^6}+\frac{432\, D\, m^4\, s^2}{(2+D)\,\triangle^6}-\frac{768\, m^4\, s}{(2+D)\,\triangle^5}-\frac{2304\, m^4\, s}{D\,(2+D)\,\triangle^5}+\frac{192\, D\, m^4\, s}{(2+D)\,\triangle^5}-\frac{2960\, m^2\, s^2}{(2+D)\,\triangle^5}+\frac{5024\, m^2\, s^2}{D\,(2+D)\,\triangle^5}+\frac{624\, D\, m^2\, s^2}{(2+D)\,\triangle^5}-\frac{40\, D^2\, m^2\, s^2}{(2+D)\,\triangle^5}+\frac{192\, m^4}{(2+D)\,\triangle^4}+\frac{96\, D\, m^4}{(2+D)\,\triangle^4}-\frac{32\, m^2\, s}{(2+D)\,\triangle^4}-\frac{1216\, m^2\, s}{D\,(2+D)\,\triangle^4}+\frac{208\, D\, m^2\, s}{(2+D)\,\triangle^4}-\frac{40\, D^2\, m^2\, s}{(2+D)\,\triangle^4}-\frac{432\, s^2}{(2+D)\,\triangle^4}+\frac{288\, s^2}{D\,(2+D)\,\triangle^4}+\frac{208\, D\, s^2}{(2+D)\,\triangle^4}-\frac{36\, D^2\, s^2}{(2+D)\,\triangle^4}+\frac{2\, D^3\, s^2}{(2+D)\,\triangle^4}+\frac{128\, m^2}{\triangle^3}-\frac{32\, D\, m^2}{\triangle^3}+\frac{112\, s}{\triangle^3}-\frac{96\, s}{D\,\triangle^3}-\frac{40\, D\, s}{\triangle^3}+\frac{4\, D^2\, s}{\triangle^3}+\frac{32}{\triangle^2}-\frac{24\, D}{\triangle^2}+\frac{4\, D^2}{\triangle^2}$$

```
In[•]:= A2full /. Δ → 1 / FAD[{k, m}];
        TID[%, k] // FeynAmpDenominatorSplit // FCE;
        A2red = % /. kinematics /. FAD[{k, m}] → 1 / Δ // Expand
```

Out[•]=
$$\frac{8960\,m^8\,s^2}{D\,(2+D)\,\Delta^8}-\frac{2304\,m^6\,s^2}{(2+D)\,\Delta^7}+\frac{21\,120\,m^6\,s^2}{D\,(2+D)\,\Delta^7}-\frac{640\,m^6\,s}{(2+D)\,\Delta^6}-\frac{1280\,m^6\,s}{D\,(2+D)\,\Delta^6}-\frac{4240\,m^4\,s^2}{(2+D)\,\Delta^6}+\frac{16\,864\,m^4\,s^2}{D\,(2+D)\,\Delta^6}+\frac{224\,D\,m^4\,s^2}{(2+D)\,\Delta^6}-\frac{832\,m^4\,s}{(2+D)\,\Delta^5}-\frac{2176\,m^4\,s}{D\,(2+D)\,\Delta^5}+\frac{128\,D\,m^4\,s}{(2+D)\,\Delta^5}-\frac{2400\,m^2\,s^2}{(2+D)\,\Delta^5}+\frac{5280\,m^2\,s^2}{D\,(2+D)\,\Delta^5}+\frac{264\,D\,m^2\,s^2}{(2+D)\,\Delta^5}+\frac{192\,m^4}{(2+D)\,\Delta^4}+\frac{96\,D\,m^4}{(2+D)\,\Delta^4}-\frac{208\,m^2\,s}{(2+D)\,\Delta^4}-\frac{992\,m^2\,s}{D\,(2+D)\,\Delta^4}+\frac{128\,D\,m^2\,s}{(2+D)\,\Delta^4}-\frac{8\,D^2\,m^2\,s}{(2+D)\,\Delta^4}-\frac{448\,s^2}{(2+D)\,\Delta^4}+\frac{576\,s^2}{D\,(2+D)\,\Delta^4}+\frac{72\,D\,s^2}{(2+D)\,\Delta^4}+\frac{8\,D^2\,s^2}{(2+D)\,\Delta^4}-\frac{2\,D^3\,s^2}{(2+D)\,\Delta^4}+\frac{128\,m^2}{\Delta^3}-\frac{32\,D\,m^2}{\Delta^3}+\frac{32\,s}{\Delta^3}-\frac{48\,s}{D\,\Delta^3}-\frac{4\,D\,s}{\Delta^3}+\frac{32}{\Delta^2}-\frac{24\,D}{\Delta^2}+\frac{4\,D^2}{\Delta^2}$$

- Perform loop integrals and expand in ϵ ($d = 4 - 2\,\epsilon$)

```
In[•]:= A1red /. Δ^n_ → master[-n];
        resA1 = Series[% /. D → 4 - 2 ϵ, {ϵ, 0, 0}] // Normal
```

Out[•]= $-\frac{19\,i\,s^2}{720\,m^4\,\pi^2}$

```
In[•]:= A2red /. Δ^n_ → master[-n];
        resA2 = Series[% /. D → 4 - 2 ϵ, {ϵ, 0, 0}] // Normal
```

Out[•]= $-\frac{11\,i\,s^2}{360\,m^4\,\pi^2}$

- Multiply by the prefactor and rewrite in terms of α (factor of 2 for identical diagrams)

```
In[•]:= A1final = 2*prefactor*resA1 /. e → Sqrt[4*Pi*α]
```

Out[•]= $\frac{38\, i\, s^2\, \alpha^2}{45\, m^4}$

```
In[•]:= A2final = 2*prefactor*resA2 /. e → Sqrt[4*Pi*α]
```

Out[•]= $\frac{44\, i\, s^2\, \alpha^2}{45\, m^4}$